# $Id: PKGBUILD 174764 2013-01-07 09:30:41Z giovanni $
# Maintainer: Giovanni Scafora <giovanni@archlinux.org>
# Contributor: Filip Dvorak <fila@pruda.com>

pkgname=enca
pkgver=1.14
pkgrel=1
pkgdesc="Charset analyser and converter"
arch=('i686' 'x86_64')
url="http://cihar.com/software/enca/"
options=('!libtool')
license=('GPL')
depends=('recode')
source=("http://dl.cihar.com/${pkgname}/${pkgname}-${pkgver}.tar.bz2")
md5sums=('03792f25aed003d3ea33b3ce1be1f256')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  ./configure --prefix=/usr \
              --libexecdir=/usr/lib \
              --mandir=/usr/share/man \
              --with-librecode=/usr \
              --enable-external
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  make DESTDIR="${pkgdir}" install
}
